*Journal of Radioanalytical Chemistry, Vol. 20 (1974) 769– 777*

# A SIMPLE PROGRAM IN "BASIC" LANGUAGE FOR ANALYSIS OF GAMMA-SPECTRA USING AN ON-LINE MINICOMPUTER

F. W. LIMA, L. T. ATALLA

*Instituto de Energia Atômica, C. P. 11049, São Paulo, SP (Brazil)*

(Received March 30, 1973)

A program in "INSTRUMENT BASIC" language is proposed for analysis of gamma-ray spectra obtained with Ge(Li) detectors and accumulated in multichannel analysers on-line with minicomputers. The program locates the peaks, evaluates the corresponding energy values, the net peak areas and the standard deviations on the areas.

## Introduction

Many advantages arise from the use of minicomputers, on-line with the analyser, for the interpretation of gamma-spectra. In the case of the preparation of programs for ultimate use in activation analysis problems, the interference of the programmer with the equipment is often necessary up to the point where all the parameters for the program have been chosen. On-line computers are extremely useful for this phase of the work. If the program language is a conversational one, such as "BASIC", the advantages of the on-line computer can be exploited at their highest capacities. "BASIC" language is also a good starting point for beginners in programming and it is powerful enough for an experienced programmer to make use of it for advanced work.

In this paper a program is presented in "INSTRUMENT BASIC"[1] language, which is a "BASIC" language adapted for use with the 5402A Hewlett-Packard System, composed of a multichannel analyser (4096 channels, Model 5401B), a 2100A computer (8 K memory, expandable to 32 K) and a teleprinter, Model HP 2752A. Tape perforation and tape reading can be done through the teleprinter, although perforation and reading will be much faster if a fast tape punch and fast tape reader are used. In our case we used a HP Model 2895A tape punch and a HP Model 2748A tape reader.

The program, which is written for spectra from Ge(Li) detectors, can cover the whole memory of the analyser or smaller parts of it, such as one half, leaving the other half for accumulation of spectra or for comparison or subtraction of background. The listing of the program contains about 1,500 characters for spectra stored in 2,048 or 4,096 channels. This size is smaller than the one proposed by Kemper and van Kempen,[2] which is also written in BASIC, for spectra accumulated in 400 channels, but involving about 4,800 characters and containing

the requirement that data are read into the computer memory in a separate file in order to avoid exceeding the maximum number of characters permitted by the computer time sharing system. The execution time for the program by Kemper and van Kempen is very fast, about 28.8 sec for a 400-channel capacity. The execution and typing time for the program presented in this paper is equal to about 4 min for a spectrum accumulated in 2,048 or 4,096 channels, corresponding to a sample composed of $^{241}Am$, $^{170}Tm$, $^{137}Cs$, $^{152\text{-}154}Eu$, $^{22}Na$, $^{57}Co$ and $^{60}Co$. Data reading is included in the program itself.

Indexed variables are avoided, in order not to put restrictions on the memory size of the computer, by having to use a DIMENSION statement that occupies an appreciable part of the computer memory. The program is conceived such that after all calculations are carried out in connection with a peak, the corresponding data are printed and erased from the computer memory, leaving it available for the following set of calculations concerning the next peak.

## Theory

The complete program is made up of two parts: the first is the calibration of the address or channels scale of the analyser in terms of energy units. The second part of the program analyses the spectra by locating the peaks, determining the centroid values for the peaks, calculating their corresponding energies in keV, the area for each peak and the associated standard deviation. A loop is introduced in the program to re-start the calculations for another sample, without having to go through the calibration steps for each new sample to be counted. However, the calibration step can be repeated as many times as one wishes by typing "STOP" after the analysis of a spectrum has been completed.

The areas of the peaks are calculated by summation of the number of counts in each channel of the peak and without assuming a Gaussian peak shape. If such an assumption is made, allowance should be made for asymmetry of the peaks. Kemper and van Kempen[2] used the same procedure for the same reason.

### Calibration

Calibration of the channels scale is made with only two points because gain and zero stability of the equipment is very good, especially if a peak stabilizer is used. A linear relation between energy and channel-numbers is assumed. Such an assumption is more close to truth the smaller the number of channels involved.

The choice of the lower and upper channel for a peak is made in such a way that the left and right boundaries for a peak do not lay in the "valley" immediately before and after a peak. Instead, these channels are chosen at the beginning of the steep and nearly linear part of the peak, on both sides. Such a choice of lower and




Fig. 1. Schematic indication of symbols for calculating net area and peak centroid (see text)

upper channels can be made, during the calibration step, by visual inspection of the oscilloscope image of the spectrum. Choosing a smaller number of channels on both sides of the centroid, rather than including all those between the centroid and valleys, has the advantage of minimizing interferences from other peaks that might exist immediately before and after the peak for which the calculation is being carried out by the computer.

However, if peak boundaries are selected close to the centroid on both sides of the peak, resulting in a very small number of channels in the peak, the value for the relative standard deviation Y/B2, will be very high (Y is the standard deviation for the net area value, as given by Eq. (7), and B2 the net area of the peak). The choice of the boundaries as the first channel before the linear part of the peak, on both sides has shown, in practice, to be a good compromise, giving a peak with a sufficient number of channels for area calculations and avoiding interferences from other peaks that are near to the one being analysed.

After choosing the lower (L) and upper (U) channels of a calibration peak (Fig. 1), the counts S1 and S2 in these channels are registered by the program. The program calculates next the slope and the intercept for the straight line, MK + B, passing through the points (L, S1) and (U, S2). The trapezoidal area defined by the points (L, Φ), (L, S1), (U, S2) and (U, Φ) is subtracted, channel by channel, from the total area of the peak. This results in the net values N for the number of counts per channel.

The summation of the N values will give the net area for the peak. The centroid K1 of a peak is calculated by the ratio of the sum of the first moment (product

of channel content N and corresponding channel number K) to the net area of the peak, for K varying from L to U:

$$K1 = \sum_{K=L}^{U} NK \Big/ \sum_{K=L}^{U} N \tag{1}$$

The ratio of the difference between the energy values for the calibration peaks and the corresponding centroid differences, gives the value R in keV per channel:

$$R = (E'' - E')/(K1'' - K1') \tag{2}$$

In general the calibration line does not pass through the origin of the graph "energy versus channels", and the point where this line intercepts the abscissa X is calculated by

$$X = K1'' - E''/R = K1' - E'/R \tag{3}$$

## Spectrum analysis

In order to detect a peak, a "peak half-width" L1, defined by the function INT, meaning "the integer value of", is associated with each peak:

$$L1 = INT[4*(1/R) + 0.5] \tag{4}$$

In this way the width is recalculated if the amplifier gain is altered by the user. Working in the energy range of zero to 2 MeV, for instance, the number of channels, 2L1, involved in a peak will be 16 for a spectrum accumulated in 4,096 channels ($R \cong 0.4$ keV/channel) or 8 for a spectrum accumulated in 2,048 channels ($R \cong 1.0$ keV/channel).

Next the program requests the interval or zone of the spectrum to be analysed, defined by channels Z1 up to Z2. The program then reads the counting values C in each channel K of this interval. A comparison is made of the counts C in channel K with the counts in channels lower and higher by two units with respect to channel K, that is (K − 2, C1) and (K + 2, C2). The condition for a peak to "exist" in channel K is:[3]

$$C > C1 + P*SQR(C1) \tag{5}$$

$$C > C2 + P*SQR(C2) \tag{6}$$

with SQR standing for "square root of".

P, in the above expressions, is a "Sensitivity Factor" used to avoid interpretation, by the program, of statistical counting fluctuations as peaks and has an empirical value between 1 and 2. In this way, only counting values C exceeding C1 and C2 by once or twice the standard deviation in these two channels are retained by the program, in order to calculate the centroid for the peak and to proceed with subsequent calculations.

Once a peak has been identified by its centroid value, the net area of that peak is calculated for the 2L1 channels in the peak. If B1 is the total uncorrected area for a peak, the standard deviation Y on the net area is calculated by

$$Y = SQR[B1 + (L1)^2 * (S1 + S2)] \tag{7}$$

A peak is retained as a true peak if the net area B2 of the peak is larger or at least equal to 2 times the standard deviation, that is,

$$B2 \geq 2 * Y \tag{8}$$

This condition, plus the ones expressed by Eqs. (5) and (6), are sufficient for the analysis of a spectrum in the interval from zero energy up to the highest energy peak in the spectrum. In the highest energy part of the spectrum, however, the number of counts of possible peaks superimposed on the Compton continuum caused by higher-energy gamma-rays existing above the chosen energy range, is often of the same order of magnitude as the statistical fluctuations in each channel of the Compton continuum or the true background. To avoid printing of data corresponding to this part of the spectrum, where peaks if they exist are of no use, a further empirical condition is imposed, stating that the net counting rate in a peak area must be at least equal to 15 counts per minutes. This would correspond to an average of one to two counts per minute and per channel in the peak, above the Compton continuum or the true background.

In order not to waste time with calculations which are not useful, such as the calculations corresponding to channels or energy values above the highest energy peak actually present in the spectrum, the Z2 value can be put equal to the channel number where the last peak of interest is located, plus 20 channels, for instance. The peak of highest energy present in the spectrum can be visually located by inspection of the spectrum on the display of the oscilloscope.

After calculation of the net area and standard deviation for a peak, these values are divided by the counting or live-time during which the spectrum was accumulated, giving the final results in counts per minute. Both values are printed out together with the channel number K1 for the centroid and its value in energy units:

$$(K1 - X) * R. \tag{9}$$

This program is not intended for precise determination of energies, since the calibration of the channel axis is made on the assumption of a linear correlation between channel number and energy, which is rigorously valid only for small channel intervals. The approximation, however, is sufficiently good for activation analysis work, where other information is available for the identification of the radioactive species, such as some previous chemical group separation, decay curves, irradiation times, etc. The approximation for the energies that one may obtain with this program can be evaluated by comparing the values shown in Table 1 and 2 with a catalog of gamma-ray energies values, such as the one prepared by Pagden et al.[4]

Table 1

Data output for "Sensitivity Factor" equal to 2

```
SPECTRUM ANALYSIS WITH CALIBRATION:KEV,CPM,ST.DEV.

CALIBRATION
RADIOISOTOPES FOR CALIBRATION ?AM241,CO60
LOWER AND UPPER CHANNEL FOR PEAK 1              ?129,142
LOWER AND UPPER CHANNEL FOR PEAK 2              ?2774,2786
ENERGIES FOR CALIBRATION PEAKS E1,E2            ?59.543,1332.49
KEV PER CHANNEL:                  .431435
ZERO ENERGY CHANNEL:              12.4331

SPECTRUM ANALYSIS
PAUSE
CONTINUE
REMARKS
?"CO57,AM241,CO60,NA22,CS137"    0
LIVE TIME T                      ?20
SENSITIVITY FACTOR P             ?2
SPECTRUM ZONE Z1,Z2              ?0,4096

  CHANNEL          KEV          AREA(CPM)        STAND.DEV.

 15.242         1.35232          25114.4          56.2648
 73.3335        31.7263          433.7            54.0512
 135.915        59.4434          13050            69.3229
 266.435        122.287          30251.3          73.1433
 296.351        136.633          3631.6           46.5303
 1174.15        511.146          5063.87          32.3747
 1337           661.763          4331.65          25.4336
 2449.23        1173.16          1372.32          15.3362
 2659.79        1274.53          1314.7           11.313
 2733.05        1332.43          1623.6           13.4437
PAUSE
```

## Program

In Tables 1 and 2 an example is presented of the application to the spectrum of a sample composed of the following radioisotopes: $^{241}Am$, $^{22}Na$, $^{137}Cs$, $^{57}Co$ and $^{60}Co$. The 59.543 keV peak for $^{241}Am$ and the 1332.49 keV peak for $^{60}Co$ were used as internal standards for calibration. In Table 1 the "Sensitivity Factor" P was set equal to 2 and in Table 2 it was set equal to 1.5. A listing of the program is presented in Table 3.

In Tables 1 and 2 the statements followed by a question mark correspond to data that should be supplied by the user. Statements followed by a colon are partial answers supplied by the computer.

The Authors are grateful to Miss Mitiko Miyamaru for her kind help.

Table 2

Data output for "Sensitivity Factor" equal to 1.5

```
ONTINUE
EMARKS
C057,AM241,C060,NA22,CS137
IVE TIME T                    ?20
ENSITIVITY FACTOR P           ?1.5
PECTRUM ZONE Z1,Z2            ?3,4096

  CHANNEL          KEV          AREA(CPM)        STD.DEV.

 5.242        1.35232          25114.4          56.2643
12.1772       14.3198          1359.45          44.3277
28.3335       31.7263          483.7            94.0513
35.915        59.4434          13458            60.3089
66.433        122.237          30251.3          76.1433
96.351        136.633          3631.6           48.5362
95.945        134.636          7085.675         36.0964
87.338        228.636          62.95            32.915
63.567        412.173          33.775           24.9253
074.15        511.146          5263.37          30.3747
275.64        603.153          27.325           16.37
367           661.763          4331.65          25.4336
353.26        686.649          25.35            15.3333
256.32        1094.73          26.325           13.545
449.23        1173.16          1672.32          15.3362
559.79        1274.53          1014.7           11.313
730.05        1332.43          1623.6           13.4457
JSE
```

Table 3

Listing of the program

```
10  PRINT "SPECTRUM ANALYSIS WITH CALIBRATION:KEV,CPM,ST.DEV."
12  PRINT
20  PRINT "CALIBRATION"
30  PRINT "RADIOISOTOPES FOR CALIBRATION",
31  INPUT 00
40  FOR P1=1 TO 2
50  LET A=E=0
60  PRINT "LOWER AND UPPER CHANNEL FOR PEAK";P1,
61  INPUT L,U
70  CALL READ(L,S1)
80  CALL READ(U,S2)
90  LET M=(S2-S1)/(U-L)
100  LET B=S1-M*L
110  FOR K=L TO U
120  CALL READ(K,D3)
130  LET N=D3-(M*K+B)
140  LET A=A+N
150  LET E=E+N*K
160  NEXT K
170  IF P1=2 THEN 200
180  LET K1=E/A
190  NEXT P1
200  LET K2=E/A
210  PRINT "ENERGIES FOR CALIBRATION PEAKS E1,E2",
211  INPUT E1,E2
220  LET R=(E2-E1)/(K2-K1)
230  PRINT "KEV PER CHANNEL:",TAB(30),R
240  LET X=K2-E2/R
250  PRINT "ZERO ENERGY CHANNEL:",X
260  PRINT
280  PRINT "SPECTRUM ANALYSIS"
282  PAUSE
285  PRINT "REMARKS"
286  INPUT 01
290  PRINT "LIVE TIME T",TAB(30),
291  INPUT T
300  PRINT "SENSITIVITY FACTOR P",
311  INPUT P
315  PRINT "SPECTRUM ZONE Z1,Z2",
316  INPUT Z1,Z2
320  PRINT
330  PRINT "   CHANNEL";"          KEV","AREA(CPM)","STAND.DE
340  PRINT
360  LET L1=INT(4*(1/R)+.5)
380  LET B3=0
400  FOR K=Z1+L1 TO Z2-L1
405  LET E=B1=B2=0
```

```
10  CALL READ(K,C)
12  CALL READ(K-2,C1)
14  CALL READ(K+2,C2)
20  IF C<C1+P*SQR(C1) THEN 635
30  IF C<C2+P*SQR(C2) THEN 635
40  CALL READ(K-L1,S1)
50  CALL READ(K+L1,S2)
60  LET M=(S2-S1)/(2*L1)
70  LET B=S1-M*(K-L1)
80  FOR J=(K-L1) TO (K+L1)
90  CALL READ(J,C)
00  LET B1=B1+C
10  LET N=C-(M*J+B)
20  LET B2=B2+N
30  LET E=E+N*J
40  NEXT J
50  LET Y=SQR(B1+(L1)↑2*(S2+S1))
60  IF B2 <= P*Y THEN 635
70  LET K1=E/B2
80  IF B3#0 THEN 630
90  LET K2=K1
00  LET B3=B2
10  LET Y1=Y
20  GOTO 680
30  IF B2>B3 THEN 655
35  IF B3/T<15 THEN 670
40  PRINT K2,(K2-X)*R,B3/T,Y1/T
50  GOTO 670
55  IF B2/T<15 THEN 670
60  PRINT K1,(K1-X)*R,B2/T,Y/T
70  LET B3=0
80  NEXT K
00  GOTO 282
10  END
```

## References


1. HP Instrument Basic for 5402A System, A Manual Prepared for Use with the HP 5402 MCA/BASIC SYSTEM, Hewlett-Packard Company, Santa Clara, California, 1971.
2. A. Kemper, G.A. van Kempen, J. Radioanal. Chem., 6 (1970) 461.
3. G. Guzzi, J. Pauly, G. Girardi, B. Dorpema, Report EUR-3469, 1967.
4. I.M.H. Pagden, G.J. Pearson, J.M. Bewers, J. Radioanal. Chem., 8 (1971) 127.